Chapter 13
INSET Statement

Chapter Table of Contents

Chapter 13
INSET Statement

Overview

The INSET statement allows you to enhance a cusum chart by adding a box or table (referred to as an *inset*) of summary statistics directly to the graph. A possible application of an inset is to present cusum parameters on the chart rather than displaying them in a legend. An inset can also display arbitrary values provided in a SAS data set.

Note that the INSET statement by itself does not produce a display but must be used in conjunction with an XCHART statement. Insets are not available with line printer output, so the INSET statement is not applicable when the LINEPRINTER option is specified in the PROC CUSUM statement.

You can use options in the INSET statement to

- specify the position of the inset
- specify a header for the inset table
- specify graphical enhancements, such as background colors, text colors, text height, text font, and drop shadows

Getting Started

This section introduces the INSET statement with a basic example showing how it is used. See Chapter 45, "INSET and INSET2 Statements," in Part 9, "The SHEWHART Procedure," for a complete description of the INSET statement.

This example is based on the same scenario as the first example in the "Getting Started" section of Chapter 12, "XCHART Statement." A machine fills cans with oil additive and a two-sided cusum chart is used to detect shifts from the target mean of 8.100 ounces. The following statements create the data set OIL and request a two-sided cusum chart with an inset:

```
data oil;
   label hour = 'Hour';
   input hour @;
   do i=1 to 4;
      input weight @;
      output;
      end;
   drop i;
   datalines;
 1  8.024  8.135  8.151  8.065
 2  7.971  8.165  8.077  8.157
 3  8.125  8.031  8.198  8.050
 4  8.123  8.107  8.154  8.095
 5  8.068  8.093  8.116  8.128
 6  8.177  8.011  8.102  8.030
 7  8.129  8.060  8.125  8.144
 8  8.072  8.010  8.097  8.153
 9  8.066  8.067  8.055  8.059
10  8.089  8.064  8.170  8.086
11  8.058  8.098  8.114  8.156
12  8.147  8.116  8.116  8.018
;


symbol v=dot;
title 'Cusum Chart for Average Weights of Cans';
proc cusum data=oil;
   xchart weight*hour /
      mu0    = 8.100          /* target mean              */
      sigma0 = 0.050          /* known standard deviation */
      delta  = 1              /* shift to be detected     */
      alpha  = 0.10           /* Type 1 error probability */
      vaxis  = -5 to 3
      nolegend;
   label weight='Cumulative Sum';
   inset arl0 alpha delta h k mu0 shift sigmas / pos = sw;
run;
```

The resulting cusum chart is shown in Figure 13.1.



Figure 13.1. Two-Sided Cusum Chart with an Inset

Syntax

The syntax for the INSET statement is as follows:

INSET *keyword-list* < **/** *options* >**;**

You can use any number of INSET statements in the CUSUM procedure. Each INSET statement produces a separate inset and must follow an XCHART statement. The inset appears on every panel (page) produced by the last XCHART statement preceding it.

Keywords specify the statistics to be displayed in an inset; options control the inset's location and appearance. A complete description of the INSET statement syntax is given starting on page 1601 of Part 9, "The SHEWHART Procedure." The INSET statement options are identical in the CUSUM and SHEWHART procedures, but the available keywords are different. The keywords available with the CUSUM procedure are listed in Table 13.1 to Table 13.3.

Table 13.1. Summary Statistics

ARL0	average run length for zero shift
ARLDELTA	average run length for shift of δ
DATA=	arbitrary values from *SAS-data-set*
N	nominal subgroup size
NMIN	minimum subgroup size
NMAX	maximum subgroup size

Table 13.2. Parameters for One-Sided (Decision Interval) Cusum Scheme

DELTA	shift to be detected as multiple of standard error
H	decision interval h as a multiple of standard error
HEADSTART	headstart value S_0 as a multiple of standard error
K	reference value k
MU0	target mean μ_0
SHIFT	shift to be detected in data units
STDDEV	estimated or specified process standard deviation

Table 13.3. Parameters for Two-Sided (V-Mask) Cusum Scheme

ALPHA	probability of Type 1 error
BETA	probability of Type 2 error
H	vertical distance between V-mask origin and upper (or lower) arm
K	slope of lower arm of V-mask
SIGMAS	probability of Type 1 error as probability that standard normally distributed variable exceeds a specified value in absolute value

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC® User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

SAS/QC® User's Guide, Version 8

ISBN 1–58025–493–4

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999